
DRAWDATA

PURPOSE
Draws a line segment in units of the most recent plot.

DESCRIPTION
The 2 pairs of coordinates define the (x,y) values for the tail and the head (respectively) of the line segment.

SYNTAX 1
DRAWDATA <x1>   <y1>   <x2>   <y2>
where <x1> is a number or parameter that specifies the x coordinate for one end of the line segment;

 <y1> is a number or parameter that specifies the y coordinate for one end of the line segment;
 <x2> is a number or parameter that specifies the x coordinate for the other end of the line segment;

and  <y2> is a number or parameter that specifies the y coordinate for the other end of the line segment.

This syntax draws from (x1,y1) to (x2,y2).

SYNTAX 2
DRAWDATA <x1>   <y1>
where <x1> is a number or parameter that specifies the x coordinate for one end of the line segment;
and  <y1> is a number or parameter that specifies the y coordinate for one end of the line segment.

This syntax draws from the current point (typically specified by a MOVE or MOVEDATA command) to (x1,y1).

SYNTAX 3
DRAWDATA <x1> <y1> <x2> <y2> ... <xn> <yn> etc.
where <x1> is a number or parameter that specifies the x coordinate for the first point of the line segment;

 <y1> is a number or parameter that specifies the y coordinate for the first point of the line segment;
 <x2> is a number or parameter that specifies the x coordinate for the second point of the line segment;
 <y2> is a number or parameter that specifies the y coordinate for the second point of the line segment.
 <xn> is a number or parameter that specifies the x coordinate for the nth point of the line segment;

 and  <yn> is a number or parameter that specifies the y coordinate for the nth point of the line segment.

This syntax draws from (x1,y1) to (x2,y2) to (x3,y3) and so on for each coordinate pair listed.

EXAMPLES
DRAWDATA 345 216 98 211
DRAWDATA X1 Y1 X2 Y2

NOTE 1
The DRAW command draws line segments in DATAPLOT screen units (i.e., 0 to 100 in both the horizontal and vertical directions). In
contrast, the DRAWDATA command uses the coordinates of the most recent plot. A typical use is to draw reference lines on a plot.

NOTE 2
The line style (i.e., solid, dash), color, and thickness of the line segment are controlled by the first entry of the LINE, LINE COLOR,
and LINE THICKNESS commands. In particular, for SYNTAX 3 (where multiple lines are drawn) each of the line segments is drawn
with the same attributes.

NOTE 3
Line segments defined by the SEGMENT COORDINATES command are drawn whenever a subsequent plot is generated. Line
segments defined by the DRAWDATA or DRAW commands are generated immediately.

DEFAULT
None

SYNONYMS
None

RELATED COMMANDS
DRAW = Draws a line in units of 0 to 100.

SEGMENT COORDINATES = Draws line segments on subsequent plots.
MOVE = Moves to a point.
ARROW = Draws an arrow.
TRIANGLE = Draws a triangle.
BOX = Draws a box.
LINE = Sets the line type for figures and plot lines.
LINE THICKNESS = Sets the line thickness for figures and plot lines.
LINE COLOR = Sets the line colors for figures and plot lines.
CROSS-HAIR = Activates and reads the cross-hair.
TEXT = Writes a text string.

APPLICATIONS
Presentation Graphics

IMPLEMENTATION DATE
94/9

PROGRAM
SKIP 25
READ SHEESLE2.DAT Y PROC PLANT SPEED SHIFT PROC
.
BAR ON; BAR WIDTH 0.5
LINES BLANK; TITLE DEX PARETO PLOT
.
MULTIPLOT 2 2; MULTIPLOT CORNER COORDINATES 0 0 100 100
Y1LABEL MEAN
LET A = MEAN Y; BAR BASE A
DEX MEAN PARETO PLOT Y PROC PLANT SPEED SHIFT
LET XFIRST = 1
LET XLAST = MAXIMUM XPLOT
LINE SOLID; DRAWDATA XFIRST A XLAST A; LINE BLANK
Y1LABEL MEDIAN
LET A = MEDIAN Y; BAR BASE A
DEX MEDIAN PARETO PLOT Y PROC PLANT SPEED SHIFT
LINE SOLID; DRAWDATA XFIRST A XLAST A; LINE BLANK
YLIMITS
Y1LABEL STANDARD DEVIATION
LET A = STANDARD DEVIATION Y
BAR BASE A
DEX SD PARETO PLOT Y PROC PLANT SPEED SHIFT
LINE SOLID; DRAWDATA XFIRST A XLAST A; LINE BLANK
Y1LABEL RANGE
LET A = RANGE Y
BAR BASE A
DEX RANGE PARETO PLOT Y PROC PLANT SPEED SHIFT
LINE SOLID; DRAWDATA XFIRST A XLAST A; LINE BLANK
END OF MULTIPLOT

